Oracle database not responding
I had a similar issue with an off the shelf three tier application where the app server would make a connection to the database when the user logged on.  The user activity was typically to log on in the morning and leave the app open on the desktop all day even if they did nothing for long periods.

The app server was not configured to use a shared pool and would commonly hit the maximum number of sessions after two or three days.  As you found the only resolution was to restart the database every night which is crazy unless the app owners say you can't change anything but you can restart it.  You may wish to verify if you are using shared connections  (pooling) and re examine your code to make sure connections are being dropped.

Here are some things you can check to see if this is a similar situation.  This answer here is also helpful.

Log onto the database and run this from sqlplus with a privileged account
--this gives the maximum amount of sessions

SELECT name, value 
  FROM v$parameter
 WHERE name = 'sessions'


--number of users currently connected even if they are not doing anything

SELECT COUNT(*)
  FROM v$session;


I created this procedure on the database and scheduled it for every five or ten minutes.  Then you can examine the file and see if you are hitting the maximum number of sessions.  To use this you must also create a directory PROCESS_LOG_DIR where the user running this has at least write permissions on.

CREATE OR REPLACE PROCEDURE COUNT_PROCESSES
IS
   f                    UTL_FILE.file_type;
   v_processes          NUMBER (10);
   v_current_logons     NUMBER (10);
   v_current_sessions   NUMBER (10);
   v_highwater          NUMBER (10);
BEGIN
   SELECT   COUNT ( * ) INTO v_processes FROM v$process;

   SELECT   VALUE
     INTO   v_current_logons
     FROM   v$sysstat
    WHERE   name = 'logons current';

   SELECT   sessions_current, sessions_highwater
     INTO   v_current_sessions, v_highwater
     FROM   v$license;

   f := UTL_FILE.fopen ('PROCESS_LOG_DIR', 'Process.txt', 'a');

   UTL_FILE.put_line (
      f,
         TO_CHAR (SYSDATE, 'DD-MON-YYYY HH24:MI:SS')
      || ','
      || v_processes
      || ','
      || v_current_logons
      || ','
      || v_current_sessions
      || ','
      || v_highwater
   );
   UTL_FILE.fclose (f);
EXCEPTION
   WHEN UTL_FILE.invalid_path
   THEN
      raise_application_error (
         -20000,
         'ERROR: Invalid path. Create directory or set UTL_FILE_DIR.'
      );
END;